This is the first of three notebooks which accompany the paper "Gravitational Waves on Kerr Black Holes II: Metric Reconstruction with Cosmological Constant" https://arxiv.org/abs/2510.07712. This notebook computes the metric perturbations in ingoing and outgoing radiation gauge $h_{\mu\nu}^{\mathrm{IRG/ORG}}$ in Boyer-Lindquist, ingoing, and outgoing coordinates in terms of the radial and angular mode functions $R_{\omega l m}^{(\pm 2)}(r)$ and $S_{\omega l m}^{(\pm 2)}(\theta)$.

---

1. Setup and Definitions

We allow subscripts on our variables and set the working directory to this notebook's path.

In[∘]:=
```
<< Notation`
Symbolize[ParsedBoxWrapper[SubscriptBox["_", "_"]]]
SetDirectory@NotebookDirectory[];
```

We define various useful symbols. This notebook almost never uses the built-in **`Conjugate`** function. Instead, we define any required complex conjugate with the variable suffix **`bar`**.

In[∘]:=
```
coord = {t, r, θ, ϕ};
Δ[r_] := r^2 - 2 M * r + a^2 - Λ/3 * r^2 * (r^2 + a^2)
Δθ = 1 + Λ/3 * a^2 * Cos[θ]^2;
Ξ = 1 + Λ/3 * a^2;
Σ = r^2 + a^2 * Cos[θ]^2;
ξ = r - ⅈ * a * Cos[θ];
ξbar = r + ⅈ * a * Cos[θ];
ωsub = {Conjugate[ω] → ωbar, Conjugate[ωbar] → ω};
K[ω_, m_] := (r^2 + a^2) * ω - a * m;
Kbar[ω_, m_] := (r^2 + a^2) * Conjugate[ω] - a * m;
Q[ω_, m_] := -a * ω * Sin[θ] + m * Csc[θ];
Qbar[ω_, m_] := -a * Conjugate[ω] * Sin[θ] + m * Csc[θ];
```

We introduce the $\mathcal{D}_n$ and $\mathcal{L}_n$ operators defined in Eqs. (2.31) and (2.32) in the paper, making explicit their dependence on ω and m. The adjoint operators $\mathcal{D}_n{}^\dagger$ and $\mathcal{L}_n{}^\dagger$ are defined with the variable suffix **`adj`**.

In[ ]:=
```
Dn[n_, ω_, m_, f_] := (D[f, r] - (I * Ξ * K[ω, m])/Δ[r] * f + n * Δ'[r]/Δ[r] * f)
Dnadj[n_, ω_, m_, f_] := (D[f, r] + (I * Ξ * K[ω, m])/Δ[r] * f + n * Δ'[r]/Δ[r] * f)
Dnbar[n_, ω_, m_, f_] := (D[f, r] + (I * Ξ * Kbar[ω, m])/Δ[r] * f + n * Δ'[r]/Δ[r] * f) /. ωsub
Dnadjbar[n_, ω_, m_, f_] := (D[f, r] - (I * Ξ * Kbar[ω, m])/Δ[r] * f + n * Δ'[r]/Δ[r] * f) /. ωsub
Ln[n_, ω_, m_, f_] := D[f, θ] + Ξ/Δθ * Q[ω, m] * f + n * (Cot[θ] + D[Δθ, θ]/(2 Δθ)) * f
Lnadj[n_, ω_, m_, f_] := D[f, θ] - Ξ/Δθ * Q[ω, m] * f + n * (Cot[θ] + D[Δθ, θ]/(2 Δθ)) * f
Lnbar[n_, ω_, m_, f_] := D[f, θ] + Ξ/Δθ * Qbar[ω, m] * f + n * (Cot[θ] + D[Δθ, θ]/(2 Δθ)) * f /. ωsub
Lnadjbar[n_, ω_, m_, f_] :=
 D[f, θ] - Ξ/Δθ * Qbar[ω, m] * f + n * (Cot[θ] + D[Δθ, θ]/(2 Δθ)) * f /. ωsub
```

We define a table of functions representing the tetrad components of the metric perturbation $h_{a b}$.

In[ ]:=
```
htetrad[t_, r_, θ_, ϕ_] :=
  {{hll[t, r, θ, ϕ], hln[t, r, θ, ϕ], hlm[t, r, θ, ϕ], hlmbar[t, r, θ, ϕ]},
   {hln[t, r, θ, ϕ], hnn[t, r, θ, ϕ], hnm[t, r, θ, ϕ], hnmbar[t, r, θ, ϕ]},
   {hlm[t, r, θ, ϕ], hnm[t, r, θ, ϕ], hmm[t, r, θ, ϕ], hmmbar[t, r, θ, ϕ]},
   {hlmbar[t, r, θ, ϕ], hnmbar[t, r, θ, ϕ],
    hmmbar[t, r, θ, ϕ], hmbarmbar[t, r, θ, ϕ]}};
```

The IRG and ORG conditions on the coordinate components of the metric perturbation are given in Eqs. (1.19) and 1.21) as $l^\mu h^{\text{IRG}}_{\mu\nu} = g^{\mu\nu} h^{\text{IRG}}_{\mu\nu} = 0$ or $n^\mu h^{\text{ORG}}_{\mu\nu} = g^{\mu\nu} h^{\text{ORG}}_{\mu\nu} = 0$. These conditions can be recast as conditions on the tetrad components: $h^{\text{IRG}}_{l\,b} = h^{\text{IRG}}_{m\,\bar{m}} = 0$ or $h^{\text{ORG}}_{n\,b} = h^{\text{ORG}}_{m\,\bar{m}} = 0$, where b indicates a generic tetrad label. These conditions are implemented via the following rules:

In[ ]:=
```
IRGtetrad = {hll → Function[{t, r, θ, ϕ}, 0],
   hln → Function[{t, r, θ, ϕ}, 0], hlm → Function[{t, r, θ, ϕ}, 0],
   hlmbar → Function[{t, r, θ, ϕ}, 0], hmmbar → Function[{t, r, θ, ϕ}, 0]};
ORGtetrad = {hln → Function[{t, r, θ, ϕ}, 0],
   hnn → Function[{t, r, θ, ϕ}, 0], hnm → Function[{t, r, θ, ϕ}, 0],
   hnmbar → Function[{t, r, θ, ϕ}, 0], hmmbar → Function[{t, r, θ, ϕ}, 0]};
```

2. Background Metric, Tetrad, and Projection to

Coordinates

Here, the background metric $g_{\mu\nu}$ and the tetrad vectors $\{l, n, m, \overline{m}\}$ are defined in Boyer-Lindquist (t, r, θ, ϕ), ingoing (v, r, θ, ψ), and outgoing (u, r, θ, ψ) coordinates. Hereafter, any object explicitly defined in one of these coordinate systems will have a variable name ending in **`BL`**, **`In`**, or **`Out`**, respectively. Tetrad vectors with upper and lower indices are denoted by **`up`** and **`down`**, respectively. In each coordinate system, the various inner products are checked to ensure agreement with Eq. (1.5). The outer products of the tetrad vectors (with lowered indices) are also defined, and the decomposition of the metric as a sum of outer products is checked to ensure agreement with Eq. (1.6).

The objects $e_\mu^a\, e_\nu^b$ in the three coordinate systems are defined as the 4 x 4 x 4 x 4 arrays **`CoordProj`**, where the first pair of indices corresponds to the tetrad indices a and b, and the second pair corresponds to the coordinate indices μ and ν. These objects let us express the coordinate components of the metric perturbation in terms of the tetrad components in accordance with $h_{\mu\nu} = e_\mu^a\, e_\nu^b\, h_{a\,b}$, where we define $e_\mu^1 = -\epsilon_g\, n_\mu$, $e_\mu^2 = -\epsilon_g\, l_\mu$, $e_\mu^3 = \epsilon_g\, m_\mu$, $e_\mu^4 = \epsilon_g\, m_\mu$.

At the end of each subsection for each coordinate system, the objects $h_{\mu\nu}^{\text{IRG}}$ and $h_{\mu\nu}^{\text{ORG}}$ are defined in terms of the tetrad components $h_{a\,b}$ as 4 x 4 matrices **`hIRG`** or **`hORG`** (with the appropriate variable suffix indicating the coordinate system). These expressions agree with Eqs. (2.47) and (2.54) in the paper for Boyer-Lindquist coordinates, Eqs. (B.4) and (B.5) in ingoing coordinates, and Eqs. (B.10) and (B.11) in outgoing coordinates.

2a. Boyer-Lindquist Coordinates

Background Metric and Tetrad Vectors

In[ ]:=

$$\text{gBL} = \epsilon g * \left\{\left\{-\frac{\left(\Delta[r] - \Delta\theta * a^2 \,\text{Sin}[\theta]^2\right)}{\Xi^2 * \Sigma}, 0, 0, -\frac{a * \text{Sin}[\theta]^2}{\Xi^2 * \Sigma}\left(-\Delta[r] + \Delta\theta * \left(r^2 + a^2\right)\right)\right\},\right.$$
$$\left\{0, \frac{\Sigma}{\Delta[r]}, 0, 0\right\}, \left\{0, 0, \frac{\Sigma}{\Delta\theta}, 0\right\}, \left\{-\frac{a * \text{Sin}[\theta]^2}{\Xi^2 * \Sigma}\left(-\Delta[r] + \Delta\theta * \left(r^2 + a^2\right)\right),\right.$$
$$\left.\left. 0, 0, \left(\frac{\Delta\theta * \left((r)^2 + a^2\right)^2 - \Delta[r] * a^2 * \text{Sin}[\theta]^2}{\Sigma * \Xi^2}\right) * \text{Sin}[\theta]^2\right\}\right\};$$

$$\text{lupBL} = \left\{\Xi * \frac{r^2 + a^2}{\Delta[r]}, 1, 0, \frac{a * \Xi}{\Delta[r]}\right\};$$

$$\text{nupBL} = \frac{1}{2 * \Sigma} * \left\{\Xi * \left(r^2 + a^2\right), -\Delta[r], 0, a * \Xi\right\};$$

$$\text{mupBL} = \frac{1}{\text{Sqrt}[2\,\Delta\theta] * \xi\text{bar}} * \{I * a * \Xi * \text{Sin}[\theta], 0, \Delta\theta, I * \Xi / \text{Sin}[\theta]\};$$

$$\text{mbarupBL} = \frac{1}{\text{Sqrt}[2\,\Delta\theta] * \xi} * \{-I * a * \Xi * \text{Sin}[\theta], 0, \Delta\theta, -I * \Xi / \text{Sin}[\theta]\};$$

```
ldownBL = gBL.lupBL // Simplify;
ndownBL = gBL.nupBL // Simplify;
mdownBL = gBL.mupBL // Simplify;
mbardownBL = gBL.mbarupBL // Simplify;
```

Checking Inner Products

In[ ]:=

```
{lupBL.gBL.lupBL, nupBL.gBL.nupBL,
  mupBL.gBL.mupBL , mbarupBL.gBL.mbarupBL } // Simplify
{lupBL.gBL.mupBL, lupBL.gBL.mbarupBL} // Simplify
{lupBL.gBL.nupBL, mupBL.gBL.mbarupBL} // Simplify
```

Out[ ]= {0, 0, 0, 0}

Out[ ]= {0, 0}

Out[ ]= $\{-\epsilon g, \epsilon g\}$

Outer Products

```
In[ ]:= lldownBL = Simplify[Outer[Times, ldownBL, ldownBL]] /. ϵg^2 → 1;
lndownBL = Simplify[Outer[Times, ldownBL, ndownBL]] /. ϵg^2 → 1;
lmdownBL = Simplify[Outer[Times, ldownBL, mdownBL]] /. ϵg^2 → 1;
lmbardownBL = Simplify[Outer[Times, ldownBL, mbardownBL]] /. ϵg^2 → 1;
nldownBL = Simplify[Outer[Times, ndownBL, ldownBL]] /. ϵg^2 → 1;
nndownBL = Simplify[Outer[Times, ndownBL, ndownBL]] /. ϵg^2 → 1;
nmdownBL = Simplify[Outer[Times, ndownBL, mdownBL]] /. ϵg^2 → 1;
nmbardownBL = Simplify[Outer[Times, ndownBL, mbardownBL]] /. ϵg^2 → 1;
mldownBL = Simplify[Outer[Times, mdownBL, ldownBL]] /. ϵg^2 → 1;
mndownBL = Simplify[Outer[Times, mdownBL, ndownBL]] /. ϵg^2 → 1;
mmdownBL = Simplify[Outer[Times, mdownBL, mdownBL]] /. ϵg^2 → 1;
mmbardownBL = Simplify[Outer[Times, mdownBL, mbardownBL]] /. ϵg^2 → 1;
mbarldownBL = Simplify[Outer[Times, mbardownBL, ldownBL]] /. ϵg^2 → 1;
mbarndownBL = Simplify[Outer[Times, mbardownBL, ndownBL]] /. ϵg^2 → 1;
mbarmdownBL = Simplify[Outer[Times, mbardownBL, mdownBL]] /. ϵg^2 → 1;
mbarmbardownBL = Simplify[Outer[Times, mbardownBL, mbardownBL]] /. ϵg^2 → 1;
```

Checking Decomposition of Background Metric

```
In[ ]:= gBL - ϵg * (- (lndownBL + nldownBL) + (mmbardownBL + mbarmdownBL)) // Simplify

Out[ ]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Projection from Tetrad Components to Coordinate Components

```
In[ ]:= CoordProjBL = {{nndownBL, nldownBL, -nmbardownBL, -nmdownBL},
    {lndownBL, lldownBL, -lmbardownBL, -lmdownBL},
    {-mbarndownBL, -mbarldownBL, mbarmbardownBL, mbarmdownBL},
    {-mndownBL, -mldownBL, mmbardownBL, mmdownBL}};
```

IRG and ORG Coordinate Components of Metric Perturbation in terms of Tetrad Components

```
In[ ]:= Clear[hll, hln, hlm, hlmbar, hnn, hnm, hnmbar, hmm, hmmbar, hmbarmbar]
```

In[ ]:=
```
hIRGBL[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjBL⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   IRGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hIRGBL[t, r, θ, ϕ]]
hORGBL[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjBL⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   ORGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hORGBL[t, r, θ, ϕ]]
```

Out[ ]//MatrixForm=

$$\left(\begin{array}{l}
\frac{9\,\mathrm{hnn}[t,r,\theta,\phi]}{(3+a^2\Lambda)^2}-\frac{3\sqrt{3}\,a\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta]}\;\mathrm{hnm}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]}{(3+a^2\Lambda)^2\,(i\,r+a\,\mathrm{Cos}[\theta])}-\frac{3\sqrt{3}\,a\,\sqrt{6+a^2\Lambda+a^2\Lambda\cdots}}{(3+a^2\cdots}\\
\frac{9\,(r^2+a^2\,\mathrm{Cos}[\theta]^2)\,\mathrm{hnn}[t,r,\theta,\phi]}{(3+a^2\Lambda)\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))}-\frac{3\sqrt{3}\,a\,(-i\,r+a\,\mathrm{Cos}[\theta])\cdots}{2\,(3+a^2\Lambda)}\\
\frac{3\,(r+i\,a\,\mathrm{Cos}[\theta])\,\mathrm{hnm}[t,r,\theta,\phi]}{(3+a^2\Lambda)\,\sqrt{2+\frac{2}{3}a^2\Lambda\,\mathrm{Cos}[\theta]^2}}+\frac{3\,(r-i\,a\,\mathrm{Cos}[\theta])\,\mathrm{hnmb}\cdots}{(3+a^2\Lambda)\,\sqrt{2+\frac{2}{3}a^2\cdots}}\\
-\frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\,\mathrm{Cos}[2\theta])\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta]}\;\mathrm{hnm}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]}{4\,(3+a^2\Lambda)^2\,(i\,r+a\,\mathrm{Cos}[\theta])}-\frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\,\mathrm{Cos}[2\theta])\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\cdots}}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta\cdots}
\end{array}\right.$$

Out[ ]//MatrixForm=

$$\left(\begin{array}{l}
\frac{(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))^2\,\mathrm{hll}[t,r,\theta,\phi]}{(3+a^2\Lambda)^2\,(a^2+2r^2+a^2\,\mathrm{Cos}[2\theta])^2}+\frac{\sqrt{3}\,a\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))\,\sqrt{6+a^2\Lambda+a^2\Lambda\cdots}}{2\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta])\,(i\,r\cdots}\\
-\frac{(a^2\,(-3+r^2\cdots}{2\,(3+\cdots}\\
-\frac{\sqrt{3}\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda)\cdots}{2\,(3+a^2\Lambda)\,(r-i\,a\,\mathrm{Cos}[\theta])\,\sqrt{6+a^2\cdots}}\\
-\frac{a\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))^2\,\mathrm{hll}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]^2}{(3+a^2\Lambda)^2\,(a^2+2r^2+a^2\,\mathrm{Cos}[2\theta])^2}-\frac{3a\,(a^2+r^2)\,(6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta])\,\mathrm{hmbarmbar}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]^2}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta])^2}-\frac{3\,a\cdots}{\cdots}
\end{array}\right.$$

2b. Ingoing Coordinates

Background Metric and Tetrad Vectors

In[•]:=
```
gIn = ϵg * {{-(Δ[r] - a^2 * Δθ * Sin[θ]^2)/(Ξ^2 * Σ), 1/Ξ, 0, -(a * Sin[θ]^2)/Ξ * (((r^2 + a^2) Δθ - Δ[r])/(Ξ * Σ))},
    {1/Ξ, 0, 0, -(a * Sin[θ]^2)/Ξ}, {0, 0, Σ/Δθ, 0}, {-(a * Sin[θ]^2)/Ξ * (((r^2 + a^2) Δθ - Δ[r])/(Ξ * Σ)),
     -(a * Sin[θ]^2)/Ξ, 0, 1/Ξ^2 * (((r^2 + a^2)^2 * Δθ - Δ[r] * a^2 * Sin[θ]^2)/Σ) * Sin[θ]^2}};
lupIn = {(2 Ξ * (r^2 + a^2))/Δ[r], 1, 0, (2 a * Ξ)/Δ[r]};
nupIn = {0, -Δ[r]/(2 * Σ), 0, 0};
mupIn = 1/(Sqrt[2 Δθ] * ξbar) * {I * a * Ξ * Sin[θ], 0, Δθ, I * Ξ / Sin[θ]};
mbarupIn = 1/(Sqrt[2 Δθ] * ξ) * {-I * a * Ξ * Sin[θ], 0, Δθ, -I * Ξ / Sin[θ]};
ldownIn = gIn.lupIn;
ndownIn = gIn.nupIn;
mdownIn = gIn.mupIn;
mbardownIn = gIn.mbarupIn;
```

Checking Inner Products

In[•]:=
```
{lupIn.gIn.lupIn, nupIn.gIn.nupIn,
   mupIn.gIn.mupIn, mbarupIn.gIn.mbarupIn } /. Λ → 0 // Simplify
{lupIn.gIn.mupIn, lupIn.gIn.mbarupIn} /. Λ → 0 // Simplify
{lupIn.gIn.nupIn, mupIn.gIn.mbarupIn} /. Λ → 0 // Simplify
```

Out[•]= {0, 0, 0, 0}

Out[•]= {0, 0}

Out[•]= {-ϵg, ϵg}

Outer Products

In[ ]:=
```
lldownIn = Simplify[Outer[Times, ldownIn, ldownIn]] /. ϵg^2 → 1;
lndownIn = Simplify[Outer[Times, ldownIn, ndownIn]] /. ϵg^2 → 1;
lmdownIn = Simplify[Outer[Times, ldownIn, mdownIn]] /. ϵg^2 → 1;
lmbardownIn = Simplify[Outer[Times, ldownIn, mbardownIn]] /. ϵg^2 → 1;
nldownIn = Simplify[Outer[Times, ndownIn, ldownIn]] /. ϵg^2 → 1;
nndownIn = Simplify[Outer[Times, ndownIn, ndownIn]] /. ϵg^2 → 1;
nmdownIn = Simplify[Outer[Times, ndownIn, mdownIn]] /. ϵg^2 → 1;
nmbardownIn = Simplify[Outer[Times, ndownIn, mbardownIn]] /. ϵg^2 → 1;
mldownIn = Simplify[Outer[Times, mdownIn, ldownIn]] /. ϵg^2 → 1;
mndownIn = Simplify[Outer[Times, mdownIn, ndownIn]] /. ϵg^2 → 1;
mmdownIn = Simplify[Outer[Times, mdownIn, mdownIn]] /. ϵg^2 → 1;
mmbardownIn = Simplify[Outer[Times, mdownIn, mbardownIn]] /. ϵg^2 → 1;
mbarldownIn = Simplify[Outer[Times, mbardownIn, ldownIn]] /. ϵg^2 → 1;
mbarndownIn = Simplify[Outer[Times, mbardownIn, ndownIn]] /. ϵg^2 → 1;
mbarmdownIn = Simplify[Outer[Times, mbardownIn, mdownIn]] /. ϵg^2 → 1;
mbarmbardownIn = Simplify[Outer[Times, mbardownIn, mbardownIn]] /. ϵg^2 → 1;
```

Checking Decomposition of Background Metric

In[ ]:=
```
gIn - ϵg * (- (lndownIn + nldownIn) + (mmbardownIn + mbarmdownIn)) // Simplify
```

Out[ ]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Projection from Tetrad Components to Coordinate Components

In[ ]:=
```
CoordProjIn = {{nndownIn, nldownIn, -nmbardownIn, -nmdownIn},
   {lndownIn, lldownIn, -lmbardownIn, -lmdownIn},
   {-mbarndownIn, -mbarldownIn, mbarmbardownIn, mbarmdownIn},
   {-mndownIn, -mldownIn, mmbardownIn, mmdownIn}};
```

IRG and ORG Coordinate Components of Metric Perturbation in terms of Tetrad Components

In[ ]:=
```
Clear[hll, hln, hlm, hlmbar, hnn, hnm, hnmbar, hmm, hmmbar, hmbarmbar]
```

In[ ]:=
```
hIRGIn[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjIn⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   IRGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hIRGIn[t, r, θ, ϕ]]
hORGIn[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjIn⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   ORGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hORGIn[t, r, θ, ϕ]]
```

Out[ ]//MatrixForm=

$$\frac{9\,\text{hnn}[t,r,\theta,\phi]}{(3+a^2\Lambda)^2} - \frac{3\sqrt{3}\,a\sqrt{6+a^2\Lambda+a^2\Lambda\,\text{Cos}[2\theta]}\;\text{hnm}[t,r,\theta,\phi]\,\text{Sin}[\theta]}{(3+a^2\Lambda)^2\,(i\,r+a\,\text{Cos}[\theta])} - \frac{3\sqrt{3}\,a\sqrt{6+a^2\Lambda+a^2\Lambda \ldots}}{(3+a^2 \ldots}$$

$$\frac{9\,(a^2+2r^2+a^2\,\text{Cos}[2\theta])\,\text{hnn}[t,r,\theta,\phi]}{(3+a^2\Lambda)\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))} - \frac{3\sqrt{3}\,a\,(-i\,r+a\,\text{Cos}[\theta])}{(3+a^2\Lambda)\,(\ldots}$$

$$\frac{3\sqrt{3}\,(r+i\,a\,\text{Cos}[\theta])\,\text{hnm}[t,r,\theta,\phi]}{(3+a^2\Lambda)\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\text{Cos}[2\theta]}} + \frac{3\sqrt{3}\,(r-i\,a\,\text{Cos}[\theta])\ldots}{(3+a^2\Lambda)\,\sqrt{6+a^2\Lambda+\ldots}}$$

$$-\frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\,\text{Cos}[2\theta])\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\text{Cos}[2\theta]}\;\text{hnm}[t,r,\theta,\phi]\,\text{Sin}[\theta]}{4\,(3+a^2\Lambda)^2\,(i\,r+a\,\text{Cos}[\theta])} - \frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\,\text{Cos}[2\theta])\,\sqrt{6+a^2\Lambda+a^2\Lambda\,\text{Cos}[2\ldots}}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\text{Cos}[\theta\ldots}$$

Out[ ]//MatrixForm=

$$\frac{(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))^2\,\text{hll}[t,r,\theta,\phi]}{4\,(3+a^2\Lambda)^2\,(r^2+a^2\,\text{Cos}[\theta]^2)^2} + \frac{\sqrt{3}\,a\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))\,\sqrt{6+a^2\Lambda+a^2\Lambda \ldots}}{2\,(3+a^2\Lambda)^2\,(-i\,r+a\,\text{Cos}[\theta])\,(i\,r \ldots}$$

$$-\frac{i\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))}{2\,(3+a^2\Lambda)\,(i\,r+a\,\text{Cos}[\theta])\,\sqrt{2+\frac{2}{3}\ldots}}$$

$$-\frac{a\,(a^2(-3+r^2\Lambda)+r\,(6M-3r+r^3\Lambda))^2\,\text{hll}[t,r,\theta,\phi]\,\text{Sin}[\theta]^2}{4\,(3+a^2\Lambda)^2\,(r^2+a^2\,\text{Cos}[\theta]^2)^2} - \frac{3a\,(a^2+r^2)\,(6+a^2\Lambda+a^2\Lambda\,\text{Cos}[2\theta])\,\text{hmbarmbar}[t,r,\theta,\phi]\,\text{Sin}[\theta]^2}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\text{Cos}[\theta])^2} - \frac{3a \ldots}{\ldots}$$

2c. Outgoing Coordinates

Background Metric and Tetrad Vectors

In[ ]:=

$$\text{gOut} = \epsilon g * \left\{\left\{-\frac{\Delta[r] - a^2 * \Delta\theta * \text{Sin}[\theta]^2}{\Xi^2 * \Sigma}, -\frac{1}{\Xi}, 0, -\frac{a * \text{Sin}[\theta]^2}{\Xi} * \left(\frac{(r^2 + a^2)\,\Delta\theta - \Delta[r]}{\Xi * \Sigma}\right)\right\}, \left\{-\frac{1}{\Xi}, 0, 0, \frac{a * \text{Sin}[\theta]^2}{\Xi}\right\}, \left\{0, 0, \frac{\Sigma}{\Delta\theta}, 0\right\}, \left\{-\frac{a * \text{Sin}[\theta]^2}{\Xi} * \left(\frac{(r^2 + a^2)\,\Delta\theta - \Delta[r]}{\Xi * \Sigma}\right), \frac{a * \text{Sin}[\theta]^2}{\Xi}, 0, \frac{1}{\Xi^2} * \left(\frac{(r^2 + a^2)^2 * \Delta\theta - \Delta[r] * a^2 * \text{Sin}[\theta]^2}{\Sigma}\right) * \text{Sin}[\theta]^2\right\}\right\};$$

$$\text{lupOut} = \{0, 1, 0, 0\};$$

$$\text{nupOut} = \frac{1}{\Sigma}\left\{\Xi * (r^2 + a^2), -\frac{\Delta[r]}{2}, 0, a * \Xi\right\};$$

$$\text{mupOut} = \frac{1}{\text{Sqrt}[2\,\Delta\theta] * \xi\text{bar}} * \{I * a * \Xi * \text{Sin}[\theta], 0, \Delta\theta, I * \Xi / \text{Sin}[\theta]\};$$

$$\text{mbarupOut} = \frac{1}{\text{Sqrt}[2\,\Delta\theta] * \xi} * \{-I * a * \Xi * \text{Sin}[\theta], 0, \Delta\theta, -I * \Xi / \text{Sin}[\theta]\};$$

```
lOutdown = gOut.lupOut;
nOutdown = gOut.nupOut;
mOutdown = gOut.mupOut;
mbarOutdown = gOut.mbarupOut;
```

Checking Inner Products

In[ ]:=

```
{lupOut.gOut.lupOut, nupOut.gOut.nupOut,
  mupOut.gOut.mupOut, mbarupOut.gOut.mbarupOut } // Simplify
{lupOut.gOut.mupOut, lupOut.gOut.mbarupOut} // Simplify
{lupOut.gOut.nupOut, mupOut.gOut.mbarupOut} // Simplify
```

Out[ ]= {0, 0, 0, 0}

Out[ ]= {0, 0}

Out[ ]= $\{-\epsilon g, \epsilon g\}$

Outer Products

In[ ]:=
```
llOutdown = Simplify[Outer[Times, lOutdown, lOutdown]] /. ϵg^2 → 1;
lnOutdown = Simplify[Outer[Times, lOutdown, nOutdown]] /. ϵg^2 → 1;
lmOutdown = Simplify[Outer[Times, lOutdown, mOutdown]] /. ϵg^2 → 1;
lmbarOutdown = Simplify[Outer[Times, lOutdown, mbarOutdown]] /. ϵg^2 → 1;
nlOutdown = Simplify[Outer[Times, nOutdown, lOutdown]] /. ϵg^2 → 1;
nnOutdown = Simplify[Outer[Times, nOutdown, nOutdown]] /. ϵg^2 → 1;
nmOutdown = Simplify[Outer[Times, nOutdown, mOutdown]] /. ϵg^2 → 1;
nmbarOutdown = Simplify[Outer[Times, nOutdown, mbarOutdown]] /. ϵg^2 → 1;
mlOutdown = Simplify[Outer[Times, mOutdown, lOutdown]] /. ϵg^2 → 1;
mnOutdown = Simplify[Outer[Times, mOutdown, nOutdown]] /. ϵg^2 → 1;
mmOutdown = Simplify[Outer[Times, mOutdown, mOutdown]] /. ϵg^2 → 1;
mmbarOutdown = Simplify[Outer[Times, mOutdown, mbarOutdown]] /. ϵg^2 → 1;
mbarlOutdown = Simplify[Outer[Times, mbarOutdown, lOutdown]] /. ϵg^2 → 1;
mbarnOutdown = Simplify[Outer[Times, mbarOutdown, nOutdown]] /. ϵg^2 → 1;
mbarmOutdown = Simplify[Outer[Times, mbarOutdown, mOutdown]] /. ϵg^2 → 1;
mbarmbarOutdown = Simplify[Outer[Times, mbardownIn, mbarOutdown]] /. ϵg^2 → 1;
```

Checking Decomposition of Background Metric

In[ ]:=
```
gOut - ϵg * (- (lnOutdown + nlOutdown) + (mmbarOutdown + mbarmOutdown)) // Simplify
```

Out[ ]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Projection from Tetrad Components to Coordinate Components

In[ ]:=
```
CoordProjOut = {{nnOutdown, nlOutdown, -nmbarOutdown, -nmOutdown},
   {lnOutdown, llOutdown, -lmbarOutdown, -lmOutdown},
   {-mbarnOutdown, -mbarlOutdown, mbarmbarOutdown, mbarmOutdown},
   {-mnOutdown, -mlOutdown, mmbarOutdown, mmOutdown}};
```

IRG and ORG Coordinate Components of Metric Perturbation in terms of Tetrad Components

In[ ]:=
```
Clear[hll, hln, hlm, hlmbar, hnn, hnm, hnmbar, hmm, hmmbar, hmbarmbar]
```

In[ ]:=
```
hIRGOut[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjOut⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   IRGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hIRGOut[t, r, θ, ϕ]]
hORGOut[t_, r_, θ_, ϕ_] := Collect[
  Sum[CoordProjOut⟦a1⟧⟦b1⟧ * htetrad[t, r, θ, ϕ]⟦a1⟧⟦b1⟧, {a1, 1, 4}, {b1, 1, 4}] /.
   ORGtetrad, Flatten[htetrad[t, r, θ, ϕ]], Simplify]
MatrixForm[hORGOut[t, r, θ, ϕ]]
```

Out[ ]//MatrixForm=

$$
\left(\begin{array}{l}
\frac{9\,\mathrm{hnn}[t,r,\theta,\phi]}{(3+a^2\Lambda)^2} - \frac{3\sqrt{3}\,a\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta]}\,\mathrm{hnm}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]}{(3+a^2\Lambda)^2\,(i\,r+a\,\mathrm{Cos}[\theta])} - \frac{3\sqrt{3}\,a\sqrt{6+a^2\Lambda+a^2\Lambda}}{(3+a^2} \\
\frac{3\,(r+i\,a\,\mathrm{Cos}[\theta])\,\mathrm{hnm}[t,r,\theta,\phi]}{(3+a^2\Lambda)\sqrt{2+\frac{2}{3}a^2\Lambda\,\mathrm{Cos}[\theta]^2}} + \frac{3\,(r-i\,a\,\mathrm{Cos}[\theta])\,\mathrm{hnmb}}{(3+a^2\Lambda)\sqrt{2+\frac{2}{3}a^2}} \\
-\frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\mathrm{Cos}[2\theta])\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta]}\,\mathrm{hnm}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]}{4\,(3+a^2\Lambda)^2\,(i\,r+a\,\mathrm{Cos}[\theta])} - \frac{3\sqrt{3}\,(-3a^2-2r^2+a^2\mathrm{Cos}[2\theta])\sqrt{6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2}}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta}
\end{array}\right.
$$

Out[ ]//MatrixForm=

$$
\left(\begin{array}{l}
\frac{(a^2(-3+r^2\Lambda)+r(6M-3r+r^3\Lambda))^2\,\mathrm{hll}[t,r,\theta,\phi]}{(3+a^2\Lambda)^2\,(a^2+2r^2+a^2\mathrm{Cos}[2\theta])^2} + \frac{\sqrt{3}\,a\,(a^2(-3+r^2\Lambda)+r(6M-3r+r^3\Lambda))\sqrt{6+a^2\Lambda+a^2\Lambda}}{2\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta])\,(i\,r} \\
-\frac{(a^2(-3+r^2}{(3+} \\
-\frac{\sqrt{3}\,(a^2(-3+r^2\Lambda)+r(6M-3r+r^3\Lambda)}{2\,(3+a^2\Lambda)\,(r-i\,a\,\mathrm{Cos}[\theta])\sqrt{6+a^2}} \\
-\frac{a\,(a^2(-3+r^2\Lambda)+r(6M-3r+r^3\Lambda))^2\,\mathrm{hll}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]^2}{(3+a^2\Lambda)^2\,(a^2+2r^2+a^2\mathrm{Cos}[2\theta])^2} - \frac{3a\,(a^2+r^2)\,(6+a^2\Lambda+a^2\Lambda\,\mathrm{Cos}[2\theta])\,\mathrm{hmbarmbar}[t,r,\theta,\phi]\,\mathrm{Sin}[\theta]^2}{4\,(3+a^2\Lambda)^2\,(-i\,r+a\,\mathrm{Cos}[\theta])^2} - 3a
\end{array}\right.
$$

3. Splitting Tetrad Components into Plus and Minus Pieces

As shown in Eqs. (2.49) and (2.56), the tetrad components of the metric perturbation in both IRG and ORG decompose as $h_{ab}(t, r, \theta, \phi) = e^{-i\omega t + i m\phi} h^{(+)}_{ab}(r, \theta) + e^{i\omega t - i m\phi} h^{(-)}_{ab}(r, \theta)$. This holds in ingoing and outgoing coordinates with the understanding that $e^{-i\omega t + i m\phi} = e^{i\omega r_* - i m r_\sharp} e^{-i\omega v + i m\psi} = e^{-i\omega r_* + i m r_\sharp} e^{-i\omega u + i m\psi}$, where r_* and $r_\sharp$ are defined in Eqs. (2.2) and (2.3). Note that since $\overline{h}_{ab} = h_{\mu\nu}\,\overline{e}^{\mu}_{a}\,\overline{e}^{\nu}_{b} = h_{\overline{a}\,\overline{b}}$, it follows that $h^{(-)}_{ab} = \overline{h}^{(+)}_{\overline{a}\,\overline{b}}$. Here we perform this decomposition of the tetrad components. This then requires us to modify the rules that implement the IRG and ORG conditions.

In[ ]:=
```
Clear[hllplus, hlnplus, hlmplus, hlmbarplus, hnnplus, hnmplus, hnmbarplus,
 hmmplus, hmmbarplus, hmbarmbarplus, hllminus, hlnminus, hlmminus, hlmbarminus,
 hnnminus, hnmminus, hnmbarminus, hmmminus, hmmbarminus, hmbarmbarminus]
```

In[ ]:=
```
hll[t_, r_, θ_, ϕ_] :=
 hllplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hllminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hln[t_, r_, θ_, ϕ_] :=
 hlnplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hlnminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hlm[t_, r_, θ_, ϕ_] :=
 hlmplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hlmminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hlmbar[t_, r_, θ_, ϕ_] := hlmbarplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] +
  hlmbarminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hnn[t_, r_, θ_, ϕ_] :=
 hnnplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hnnminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hnm[t_, r_, θ_, ϕ_] :=
 hnmplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hnmminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hnmbar[t_, r_, θ_, ϕ_] := hnmbarplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] +
  hnmbarminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hmm[t_, r_, θ_, ϕ_] :=
 hmmplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] + hmmminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hmmbar[t_, r_, θ_, ϕ_] := hmmbarplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] +
  hmmbarminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
hmbarmbar[t_, r_, θ_, ϕ_] := hmbarmbarplus[r, θ] * Exp[-i * ω * t + i * m * ϕ] +
  hmbarmbarminus[r, θ] * Exp[i * ωbar * t - i * m * ϕ]
```

In[ ]:=
```
IRGtetrad = {hllplus → Function[{r, θ}, 0], hlnplus → Function[{r, θ}, 0],
   hlmplus → Function[{r, θ}, 0], hlmbarplus → Function[{r, θ}, 0],
   hmmbarminus → Function[{r, θ}, 0], hllminus → Function[{r, θ}, 0],
   hlnminus → Function[{r, θ}, 0], hlmminus → Function[{r, θ}, 0],
   hlmbarminus → Function[{r, θ}, 0], hmmbarplus → Function[{r, θ}, 0]};
ORGtetrad = {hlnplus → Function[{r, θ}, 0], hnnplus → Function[{r, θ}, 0],
   hnmplus → Function[{r, θ}, 0], hnmbarplus → Function[{r, θ}, 0],
   hmmbarplus → Function[{r, θ}, 0], hlnminus → Function[{θ, ϕ}, 0],
   hnnminus → Function[{r, θ}, 0], hnmminus → Function[{r, θ}, 0],
   hnmbarminus → Function[{r, θ}, 0], hmmbarminus → Function[{r, θ}, 0]};
```

4. Splitting Coordinate Components into $\mathcal{H}$ and $\overline{\mathcal{H}}$ Pieces

A similar decomposition exists for the coordinate components of the metric perturbation $h_{\mu\nu}(t, r, \theta, \phi) = e^{-i\,\omega t + i\,m\,\phi} H_{\mu\nu}(r, \theta) + e^{i\,\omega t - i\,m\,\phi}\,\overline{H}_{\mu\nu}(r, \theta)$. Note that since the coordinate components are always real, unlike the tetrad components, so this decomposition is simply the sum of a single mode and its complex conjugate. Here, we perform this decomposition of the coordinate components in the three different coordinate systems and two radiation gauges. One can use the `CoordProj` objects defined in the previous section to show that the following relations hold: $H_{\mu\nu} = e^a_\mu\, e^b_\nu\, h^{(+)}_{a\,b}$ and

$$\overline{H}_{\mu\nu} = e_\mu^a\, e_\nu^b\, h_{a\,b}^{(-)}.$$

4a. Boyer-Lindquist Coordinates

In[ ]:=
```
ℋIRGBL[r_, θ_] := Coefficient[hIRGBL[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarIRGBL[r_, θ_] := Coefficient[hIRGBL[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
ℋORGBL[r_, θ_] := Coefficient[hORGBL[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarORGBL[r_, θ_] := Coefficient[hORGBL[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
```

4b. Ingoing Coordinates

In[ ]:=
```
ℋIRGIn[r_, θ_] := Coefficient[hIRGIn[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarIRGIn[r_, θ_] := Coefficient[hIRGIn[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
ℋORGIn[r_, θ_] := Coefficient[hORGIn[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarORGIn[r_, θ_] := Coefficient[hORGIn[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
```

4c. Outgoing Coordinates

In[ ]:=
```
ℋIRGOut[r_, θ_] := Coefficient[hIRGOut[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarIRGOut[r_, θ_] := Coefficient[hIRGOut[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
ℋORGOut[r_, θ_] := Coefficient[hORGOut[t, r, θ, ϕ], ⅇ^(ⅈ m ϕ-ⅈ t ω)]
ℋbarORGOut[r_, θ_] := Coefficient[hORGOut[t, r, θ, ϕ], ⅇ^(-ⅈ m ϕ+ⅈ t ωbar)]
```

5. Splitting Plus and Minus Pieces of Tetrad Components into A and B Pieces

Each $h_{a\,b}^{(\pm)}$ piece of the tetrad components can be further split into A and B pieces: $h_{a\,b}^{(+)} = A\, h_{a\,b}^{(+)A} + \overline{B}\, h_{a\,b}^{(+)B}$ and $h_{a\,b}^{(-)} = \overline{A}\, h_{a\,b}^{(+)A} + B\, h_{a\,b}^{(-)B}$, where A and B are the coefficients of the modes of the Hertz potentials in Eq. (3.11). (Here, we suppress the IRG and ORG superscripts.) We now perform this splitting. These A and B pieces will be set equal to some $H_{\omega l\,m}^{a\,b}$ function (or zero) once we pick a gauge below. We again must modify the rules that implement the IRG and ORG conditions.

In[ ]:=
```
Clear[hllplusA, hlnplusA, hlmplusA, hlmbarplusA, hnnplusA, hnmplusA,
  hnmbarplusA, hmmplusA, hmmbarplusA, hmbarmbarplusA, hllminusA,
  hlnminusA, hlmminusA, hlmbarminusA, hnnminusA, hnmminusA, hnmbarminusA,
  hmmminusA, hmmbarminusA, hmbarmbarminusA, hllplusB, hlnplusB, hlmplusB,
  hlmbarplusB, hnnplusB, hnmplusB, hnmbarplusB, hmmplusB, hmmbarplusB,
  hmbarmbarplusB, hllminusB, hlnminusB, hlmminusB, hlmbarminusB, hnnminusB,
  hnmminusB, hnmbarminusB, hmmminusB, hmmbarminusB, hmbarmbarminusB]
```

In[ ]:=
```
hllplus[r_, θ_] := A * hllplusA[r, θ] + Bbar * hllplusB[r, θ]
hllminus[r_, θ_] := Abar * hllminusA[r, θ] + B * hllminusB[r, θ]
hlnplus[r_, θ_] := A * hlnplusA[r, θ] + Bbar * hlnplusB[r, θ]
hlnminus[r_, θ_] := Abar * hlnminusA[r, θ] + B * hlnminusB[r, θ]
hlmplus[r_, θ_] := A * hlmplusA[r, θ] + Bbar * hlmplusB[r, θ]
hlmminus[r_, θ_] := Abar * hlmminusA[r, θ] + B * hlmminusB[r, θ]
hlmbarplus[r_, θ_] := A * hlmbarplusA[r, θ] + Bbar * hlmbarplusB[r, θ]
hlmbarminus[r_, θ_] := Abar * hlmbarminusA[r, θ] + B * hlmbarminusB[r, θ]
hnnplus[r_, θ_] := A * hnnplusA[r, θ] + Bbar * hnnplusB[r, θ]
hnnminus[r_, θ_] := Abar * hnnminusA[r, θ] + B * hnnminusB[r, θ]
hnmplus[r_, θ_] := A * hnmplusA[r, θ] + Bbar * hnmplusB[r, θ]
hnmminus[r_, θ_] := Abar * hnmminusA[r, θ] + B * hnmminusB[r, θ]
hnmbarplus[r_, θ_] := A * hnmbarplusA[r, θ] + Bbar * hnmbarplusB[r, θ]
hnmbarminus[r_, θ_] := Abar * hnmbarminusA[r, θ] + B * hnmbarminusB[r, θ]
hmmplus[r_, θ_] := A * hmmplusA[r, θ] + Bbar * hmmplusB[r, θ]
hmmminus[r_, θ_] := Abar * hmmminusA[r, θ] + B * hmmminusB[r, θ]
hmmbarplus[r_, θ_] := A * hmmbarplusA[r, θ] + Bbar * hmmbarplusB[r, θ]
hmmbarminus[r_, θ_] := Abar * hmmbarminusA[r, θ] + B * hmmbarminusB[r, θ]
hmbarmbarplus[r_, θ_] := A * hmbarmbarplusA[r, θ] + Bbar * hmbarmbarplusB[r, θ]
hmbarmbarminus[r_, θ_] := Abar * hmbarmbarminusA[r, θ] + B * hmbarmbarminusB[r, θ]
```

```
In[ ]:= IRGtetrad = {hllplusA → Function[{r, θ}, 0], hlnplusA → Function[{r, θ}, 0],
    hlmplusA → Function[{r, θ}, 0], hlmbarplusA → Function[{r, θ}, 0],
    hmmbarminusA → Function[{r, θ}, 0], hllminusA → Function[{r, θ}, 0],
    hlnminusA → Function[{r, θ}, 0], hlmminusA → Function[{r, θ}, 0],
    hlmbarminusA → Function[{r, θ}, 0], hmmbarplusA → Function[{r, θ}, 0],
    hllplusB → Function[{r, θ}, 0], hlnplusB → Function[{r, θ}, 0],
    hlmplusB → Function[{r, θ}, 0], hlmbarplusB → Function[{r, θ}, 0],
    hmmbarminusB → Function[{r, θ}, 0], hllminusB → Function[{r, θ}, 0],
    hlnminusB → Function[{r, θ}, 0], hlmminusB → Function[{r, θ}, 0],
    hlmbarminusB → Function[{r, θ}, 0], hmmbarplusB → Function[{r, θ}, 0]};
ORGtetrad = {hlnplusA → Function[{r, θ}, 0], hnnplusA → Function[{r, θ}, 0],
    hnmplusA → Function[{r, θ}, 0], hnmbarplusA → Function[{r, θ}, 0],
    hmmbarplusA → Function[{r, θ}, 0], hlnminusA → Function[{θ, ϕ}, 0],
    hnnminusA → Function[{r, θ}, 0], hnmminusA → Function[{r, θ}, 0],
    hnmbarminusA → Function[{r, θ}, 0], hmmbarminusA → Function[{r, θ}, 0],
    hlnplusB → Function[{r, θ}, 0], hnnplusB → Function[{r, θ}, 0],
    hnmplusB → Function[{r, θ}, 0], hnmbarplusB → Function[{r, θ}, 0],
    hmmbarplusB → Function[{r, θ}, 0], hlnminusB → Function[{θ, ϕ}, 0],
    hnnminusB → Function[{r, θ}, 0], hnmminusB → Function[{r, θ}, 0],
    hnmbarminusB → Function[{r, θ}, 0], hmmbarminusB → Function[{r, θ}, 0]};
```

6. $\mathcal{H}$ and $\overline{\mathcal{H}}$ Pieces of Coordinate Components Split into A and B Pieces

The $H_{\mu\nu}$ and $\overline{H}_{\mu\nu}$ pieces of the coordinate components can also be further split: $H_{\mu\nu} = A\,H^{A}_{\mu\nu} + \overline{B}\,H^{B}_{\mu\nu}$. Here we perform this further splitting in the three coordinate systems and the two gauges. One can use the `CoordProj` objects to show that the following relations hold: $H^{A}_{\mu\nu} = e^{a}_{\mu}\,e^{b}_{\nu}\,h^{(+)A}_{a\,b}$, $\overline{H}^{A}_{\mu\nu} = e^{a}_{\mu}\,e^{b}_{\nu}\,h^{(-)A}_{a\,b}$, and likewise for the B pieces.

6a. Boyer-Lindquist Coordinates

```
In[ ]:= ℋIRGBLA[r_, θ_] := ℋIRGBL[r, θ] /. A → 1 /. Bbar → 0
ℋIRGBLB[r_, θ_] := ℋIRGBL[r, θ] /. A → 0 /. Bbar → 1
ℋORGBLA[r_, θ_] := ℋORGBL[r, θ] /. A → 1 /. Bbar → 0
ℋORGBLB[r_, θ_] := ℋORGBL[r, θ] /. A → 0 /. Bbar → 1
ℋbarIRGBLA[r_, θ_] := ℋbarIRGBL[r, θ] /. Abar → 1 /. B → 0
ℋbarIRGBLB[r_, θ_] := ℋbarIRGBL[r, θ] /. Abar → 0 /. B → 1
ℋbarORGBLA[r_, θ_] := ℋbarORGBL[r, θ] /. Abar → 1 /. B → 0
ℋbarORGBLB[r_, θ_] := ℋbarORGBL[r, θ] /. Abar → 0 /. B → 1
```

6b. Ingoing Coordinates

In[∘]:=
```
ℋIRGInA[r_, θ_] := ℋIRGIn[r, θ] /. A → 1 /. Bbar → 0
ℋIRGInB[r_, θ_] := ℋIRGIn[r, θ] /. A → 0 /. Bbar → 1
ℋORGInA[r_, θ_] := ℋORGIn[r, θ] /. A → 1 /. Bbar → 0
ℋORGInB[r_, θ_] := ℋORGIn[r, θ] /. A → 0 /. Bbar → 1
ℋbarIRGInA[r_, θ_] := ℋbarIRGIn[r, θ] /. Abar → 1 /. B → 0
ℋbarIRGInB[r_, θ_] := ℋbarIRGIn[r, θ] /. Abar → 0 /. B → 1
ℋbarORGInA[r_, θ_] := ℋbarORGIn[r, θ] /. Abar → 1 /. B → 0
ℋbarORGInB[r_, θ_] := ℋbarORGIn[r, θ] /. Abar → 0 /. B → 1
```

6c. Outgoing Coordinates

In[∘]:=
```
ℋIRGOutA[r_, θ_] := ℋIRGOut[r, θ] /. A → 1 /. Bbar → 0
ℋIRGOutB[r_, θ_] := ℋIRGOut[r, θ] /. A → 0 /. Bbar → 1
ℋORGOutA[r_, θ_] := ℋORGOut[r, θ] /. A → 1 /. Bbar → 0
ℋORGOutB[r_, θ_] := ℋORGOut[r, θ] /. A → 0 /. Bbar → 1
ℋbarIRGOutA[r_, θ_] := ℋbarIRGOut[r, θ] /. Abar → 1 /. B → 0
ℋbarIRGOutB[r_, θ_] := ℋbarIRGOut[r, θ] /. Abar → 0 /. B → 1
ℋbarORGOutA[r_, θ_] := ℋbarORGOut[r, θ] /. Abar → 1 /. B → 0
ℋbarORGOutB[r_, θ_] := ℋbarORGOut[r, θ] /. Abar → 0 /. B → 1
```

---

7. A and B Pieces of Tetrad Components in Terms of H Functions, H Functions in Terms of Radial and Angular Mode Functions

In agreement with Eqs. (2.50) and (2.57), in both IRG or ORG, each $h_{a\,b}^{(\pm)\,A/B}$ piece equals either zero or one of the $H_{\omega l\,m}^{a\,b}$ functions. These functions are defined in Eq. (2.51) and (2.58) in terms of derivatives of the radial and angular mode functions $R_{\omega l\,m}^{(\pm 2)}(r)$ and $S_{\omega l\,m}^{(\pm 2)}(\theta)$. Here we implement this. We first express each $h_{a\,b}^{(\pm)\,A/B}$ piece in terms of the $H_{\omega l\,m}^{a\,b}$ functions and then export the variables **`hIRG`** or **`hORG`** (with the appropriate variable suffix indicating the coordinate system) as the “compact” forms of $h_{\mu\nu}^{\text{IRG/ORG}}$ in each coordinate system. We then express the $H_{\omega l\,m}^{a\,b}$ functions in terms of $R_{\omega l\,m}^{(\pm 2)}(r)$ and $S_{\omega l\,m}^{(\pm 2)}(\theta)$ and again export the variables **`hIRG`** or **`hORG`**. Finally, we export the tetrad components $h_{a\,b}^{\text{IRG/ORG}}$ in terms of $R_{\omega l\,m}^{(\pm 2)}(r)$ and $S_{\omega l\,m}^{(\pm 2)}(\theta)$, which will be used for checking the Einstein equations in the “Einstein Checks” notebook.

We define lists containing the radial and angular functions to be used in later simplifications.

In[ ]:=
```
{R[s, ω, l, m, r], R[s, -ωbar, l, -m, r],
   Rbar[s, ω, l, m, r], Rbar[s, -ωbar, l, -m, r]};
Table[%, {s, {2, -2}}];
Rlist = Flatten[{%, D[%, r], D[%, {r, 2}]}];
{S[s, ω, l, m, θ], S[s, -ωbar, l, -m, θ],
   Sbar[s, ω, l, m, θ], Sbar[s, -ωbar, l, -m, θ]};
Table[%, {s, {2, -2}}];
Slist = Flatten[{%, D[%, θ], D[%, {θ, 2}]}];
```

7a. IRG

$h_{a\,b}^{(\pm)\,A/B}$ in Terms of $H_{\omega l\,m}^{a\,b}$

In[ ]:=
```
Clear[hllplusA, hlnplusA, hlmplusA, hlmbarplusA, hnnplusA, hnmplusA,
  hnmbarplusA, hmmplusA, hmmbarplusA, hmbarmbarplusA, hllminusA,
  hlnminusA, hlmminusA, hlmbarminusA, hnnminusA, hnmminusA, hnmbarminusA,
  hmmminusA, hmmbarminusA, hmbarmbarminusA, hllplusB, hlnplusB, hlmplusB,
  hlmbarplusB, hnnplusB, hnmplusB, hnmbarplusB, hmmplusB, hmmbarplusB,
  hmbarmbarplusB, hllminusB, hlnminusB, hlmminusB, hlmbarminusB, hnnminusB,
  hnmminusB, hnmbarminusB, hmmminusB, hmmbarminusB, hmbarmbarminusB]
Clear[HIRGnn, HIRGbarnn, HIRGnm, HIRGbarnm, HIRGmm, HIRGbarmm]
```

In[ ]:=
```
hnnplusA[r_, θ_] := HIRGnn[ω, l, m, r, θ]
hnnplusB[r_, θ_] := HIRGbarnn[-ωbar, l, -m, r, θ]
hnnminusA[r_, θ_] := HIRGbarnn[ω, l, m, r, θ]
hnnminusB[r_, θ_] := HIRGnn[-ωbar, l, -m, r, θ]
hnmplusA[r_, θ_] := 0
hnmplusB[r_, θ_] := HIRGbarnm[-ωbar, l, -m, r, θ]
hnmminusA[r_, θ_] := HIRGbarnm[ω, l, m, r, θ]
hnmminusB[r_, θ_] := 0
hnmbarplusA[r_, θ_] := HIRGnm[ω, l, m, r, θ]
hnmbarplusB[r_, θ_] := 0
hnmbarminusA[r_, θ_] := 0
hnmbarminusB[r_, θ_] := HIRGnm[-ωbar, l, -m, r, θ]
hmmplusA[r_, θ_] := 0
hmmplusB[r_, θ_] := HIRGbarmm[-ωbar, l, -m, r, θ]
hmmminusA[r_, θ_] := HIRGbarmm[ω, l, m, r, θ]
hmmminusB[r_, θ_] := 0
hmbarmbarplusA[r_, θ_] := HIRGmm[ω, l, m, r, θ]
hmbarmbarplusB[r_, θ_] := 0
hmbarmbarminusA[r_, θ_] := 0
hmbarmbarminusB[r_, θ_] := HIRGmm[-ωbar, l, -m, r, θ]
```

We export $h_{\mu\nu}^{\text{IRG}}$ in terms of $H_{\omega l m}^{a\,b}$.

In[ ]:=
```
hIRGBLCompactExp = hIRGBL[t, r, θ, ϕ];
hIRGInCompactExp = hIRGIn[t, r, θ, ϕ];
hIRGOutCompactExp = hIRGOut[t, r, θ, ϕ];
DumpSave["./metric-perturbations/hIRGCompactExpressions.mx",
  {hIRGBLCompactExp, hIRGInCompactExp, hIRGOutCompactExp}];
```

$H_{\omega l m}^{a\,b}$ in terms of $R_{\omega l m}^{(-2)}(r)$ and $S_{\omega l m}^{(-2)}(\theta)$

```
In[•]:= HIRGnn[ω_, l_, m_, r_, θ_] :=
  - ϵg/4 * 1/ϱbar^2 (Sqrt[Δθ] * Lnadj[1, ω, m, Sqrt[Δθ] * Lnadj[2, ω, m, S[-2, ω, l, m, θ]]] -
      (2 I * a)/ϱ * Sin[θ] * Δθ * Lnadj[2, ω, m, S[-2, ω, l, m, θ]]) * R[-2, ω, l, m, r]
  HIRGbarnn[ω_, l_, m_, r_, θ_] := - ϵg/4 * 1/ϱ^2
     (Sqrt[Δθ] * Lnadjbar[1, ω, m, Sqrt[Δθ] * Lnadjbar[2, ω, m, Sbar[-2, ω, l, m, θ]]] +
       (2 I * a)/ϱbar * Sin[θ] * Δθ * Lnadjbar[2, ω, m, Sbar[-2, ω, l, m, θ]]) *
     Rbar[-2, ω, l, m, r] /. ωsub
  HIRGnm[ω_, l_, m_, r_, θ_] := - ϵg/(2 Sqrt[2]) * 1/ϱbar *
    (Dn[0, ω, m, Sqrt[Δθ] * Lnadj[2, ω, m, R[-2, ω, l, m, r] * S[-2, ω, l, m, θ]]] +
      (a^2 * Sin[2 θ])/Σ * Sqrt[Δθ] * Dn[0, ω, m, R[-2, ω, l, m, r]] * S[-2, ω, l, m, θ] -
      (2 * r)/Σ * Sqrt[Δθ] * Lnadj[2, ω, m, S[-2, ω, l, m, θ]] * R[-2, ω, l, m, r])
  HIRGbarnm[ω_, l_, m_, r_, θ_] :=
   - ϵg/(2 Sqrt[2]) * 1/ϱ * (Dnbar[0, ω, m, Sqrt[Δθ] * Lnadjbar[2, ω, m,
           Rbar[-2, ω, l, m, r] * Sbar[-2, ω, l, m, θ]]] + (a^2 * Sin[2 θ])/Σ * Sqrt[Δθ] *
         Dnbar[0, ω, m, Rbar[-2, ω, l, m, r]] * Sbar[-2, ω, l, m, θ] - (2 * r)/Σ * Sqrt[Δθ] *
         Lnadjbar[2, ω, m, Sbar[-2, ω, l, m, θ]] * Rbar[-2, ω, l, m, r]) /. ωsub
  HIRGmm[ω_, l_, m_, r_, θ_] :=
   - ϵg/2 * (Dn[0, ω, m, Dn[0, ω, m, R[-2, ω, l, m, r] * S[-2, ω, l, m, θ]]] -
      2/ϱ Dn[0, ω, m, R[-2, ω, l, m, r] * S[-2, ω, l, m, θ]])
  HIRGbarmm[ω_, l_, m_, r_, θ_] :=
   - ϵg/2 * (Dnbar[0, ω, m, Dnbar[0, ω, m, Rbar[-2, ω, l, m, r] * Sbar[-2, ω, l, m, θ]]] -
       2/ϱbar Dnbar[0, ω, m, Rbar[-2, ω, l, m, r] * Sbar[-2, ω, l, m, θ]]) /. ωsub
```

We export $h_{\mu\nu}^{\mathrm{IRG}}$ in terms of $R_{\omega l m}^{(-2)}(r)$ and $S_{\omega l m}^{(-2)}(\theta)$.

```
In[]:= hIRGBLExp = hIRGBL[t, r, θ, ϕ];
hIRGInExp = hIRGIn[t, r, θ, ϕ];
hIRGOutExp = hIRGOut[t, r, θ, ϕ];
DumpSave["./metric-perturbations/hIRGExpressions.mx",
  {hIRGBLExp, hIRGInExp, hIRGOutExp}];
```

We export $h_{a\,b}^{\mathrm{IRG}}$ in terms of $R_{\omega l\,m}^{(-2)}(r)$ and $S_{\omega l\,m}^{(-2)}(\theta)$.

```
In[]:= starttime = AbsoluteTime[];
{
  {0, 0, 0, 0},
  {0, hnn[t, r, θ, ϕ], hnm[t, r, θ, ϕ], hnmbar[t, r, θ, ϕ]},
  {0, hnm[t, r, θ, ϕ], hmm[t, r, θ, ϕ], 0},
  {0, hnmbar[t, r, θ, ϕ], 0, hmbarmbar[t, r, θ, ϕ]}
 };
habIRG = Collect[%, {ⅇ^(-ⅈ m ϕ+ⅈ t ωbar), ⅇ^(ⅈ m ϕ-ⅈ*ω*t)}, Collect[#,
    {A, B, Abar, Bbar}, Collect[#, Rlist, Collect[#, Slist, Simplify] &] &] &];
DumpSave["./metric-perturbations/habIRG.mx", habIRG];
AbsoluteTime[] - starttime

Out[]= 3.294548
```

7b. ORG

$h_{a\,b}^{(\pm)\,A/B}$ in Terms of $H_{\omega l\,m}^{a\,b}$

```
In[]:= Clear[hllplusA, hlnplusA, hlmplusA, hlmbarplusA, hnnplusA, hnmplusA,
  hnmbarplusA, hmmplusA, hmmbarplusA, hmbarmbarplusA, hllminusA,
  hlnminusA, hlmminusA, hlmbarminusA, hnnminusA, hnmminusA, hnmbarminusA,
  hmmminusA, hmmbarminusA, hmbarmbarminusA, hllplusB, hlnplusB, hlmplusB,
  hlmbarplusB, hnnplusB, hnmplusB, hnmbarplusB, hmmplusB, hmmbarplusB,
  hmbarmbarplusB, hllminusB, hlnminusB, hlmminusB, hlmbarminusB, hnnminusB,
  hnmminusB, hnmbarminusB, hmmminusB, hmmbarminusB, hmbarmbarminusB]
Clear[HORG_ll, HORGbar_ll, HORG_lm, HORGbar_lm, HORG_mm, HORGbar_mm]
```

In[ ]:=
```
hllplusA[r_, θ_] := HORG_ll[ω, l, m, r, θ]
hllplusB[r_, θ_] := HORGbar_ll[-ωbar, l, -m, r, θ]
hllminusA[r_, θ_] := HORGbar_ll[ω, l, m, r, θ]
hllminusB[r_, θ_] := HORG_ll[-ωbar, l, -m, r, θ]
hlmplusA[r_, θ_] := HORG_lm[ω, l, m, r, θ]
hlmplusB[r_, θ_] := 0
hlmminusA[r_, θ_] := 0
hlmminusB[r_, θ_] := HORG_lm[-ωbar, l, -m, r, θ]
hlmbarplusA[r_, θ_] := 0
hlmbarplusB[r_, θ_] := HORGbar_lm[-ωbar, l, -m, r, θ]
hlmbarminusA[r_, θ_] := HORGbar_lm[ω, l, m, r, θ]
hlmbarminusB[r_, θ_] := 0
hmmplusA[r_, θ_] := HORG_mm[ω, l, m, r, θ]
hmmplusB[r_, θ_] := 0
hmmminusA[r_, θ_] := 0
hmmminusB[r_, θ_] := HORG_mm[-ωbar, l, -m, r, θ]
hmbarmbarplusA[r_, θ_] := 0
hmbarmbarplusB[r_, θ_] := HORGbar_mm[-ωbar, l, -m, r, θ]
hmbarmbarminusA[r_, θ_] := HORGbar_mm[ω, l, m, r, θ]
hmbarmbarminusB[r_, θ_] := 0
```

We export $h_{\mu\nu}^{\text{ORG}}$ in terms of $H_{\omega l m}^{a\,b}$.

In[ ]:=
```
hORGBLCompactExp = hORGBL[t, r, θ, ϕ];
hORGInCompactExp = hORGIn[t, r, θ, ϕ];
hORGOutCompactExp = hORGOut[t, r, θ, ϕ];
DumpSave["./metric-perturbations/hORGCompactExpressions.mx",
  {hORGBLCompactExp, hORGInCompactExp, hORGOutCompactExp}];
```

$H_{\omega l m}^{a b}$ in terms of $R_{\omega l m}^{(+2)}(r)$ and $S_{\omega l m}^{(+2)}(\theta)$

```
In[]:= HORG_ll[ω_, l_, m_, r_, θ_] :=
  - ϵg/4 ξ^2 * (Sqrt[Δθ] * Ln[1, ω, m, Sqrt[Δθ] * Ln[2, ω, m, S[2, ω, l, m, θ]]] -
      (2 I * a)/ξ * Sin[θ] * Δθ * Ln[2, ω, m, S[2, ω, l, m, θ]]) * R[2, ω, l, m, r]
HORGbar_ll[ω_, l_, m_, r_, θ_] := - ϵg/4 ξbar^2 *
   (Sqrt[Δθ] * Lnbar[1, ω, m, Sqrt[Δθ] * Lnbar[2, ω, m, Sbar[2, ω, l, m, θ]]] + (2 I * a)/ξbar *
       Sin[θ] * Δθ * Lnbar[2, ω, m, Sbar[2, ω, l, m, θ]]) * Rbar[2, ω, l, m, r] /. ωsub
HORG_lm[ω_, l_, m_, r_, θ_] := ϵg/(4 * Sqrt[2]) * ξ^2/ξbar * 1/Δ[r]
   (Dnadj[0, ω, m, Sqrt[Δθ] * Ln[2, ω, m, Δ[r]^2 * R[2, ω, l, m, r] * S[2, ω, l, m, θ]]] +
     (a^2 * Sin[2 θ])/Σ * Sqrt[Δθ] * Dnadj[0, ω, m, Δ[r]^2 * R[2, ω, l, m, r] * S[2, ω, l, m, θ]] -
     (2 * r)/Σ * Sqrt[Δθ] * Ln[2, ω, m, Δ[r]^2 * R[2, ω, l, m, r] * S[2, ω, l, m, θ]])
HORGbar_lm[ω_, l_, m_, r_, θ_] :=
  ϵg/(4 * Sqrt[2]) * ξbar^2/ξ * 1/Δ[r] * (Dnadjbar[0, ω, m, Sqrt[Δθ] * Lnbar[2, ω, m,
          Δ[r]^2 * Rbar[2, ω, l, m, r] * Sbar[2, ω, l, m, θ]]] + (a^2 * Sin[2 θ])/Σ * Sqrt[Δθ] *
       Dnadjbar[0, ω, m, Δ[r]^2 * Rbar[2, ω, l, m, r] * Sbar[2, ω, l, m, θ]] - (2 * r)/Σ * Sqrt[
          Δθ] * Lnbar[2, ω, m, (Δ[r]^2) * Rbar[2, ω, l, m, r] * Sbar[2, ω, l, m, θ]]) /. ωsub
HORG_mm[ω_, l_, m_, r_, θ_] :=
  - ϵg/8 * ξ^2/ξbar^2 * (Dnadj[0, ω, m, Dnadj[0, ω, m, Δ[r]^2 * R[2, ω, l, m, r]] * S[2, ω, l, m, θ]] -
     2/ξ * Dnadj[0, ω, m, Δ[r]^2 * R[2, ω, l, m, r] * S[2, ω, l, m, θ]])
HORGbar_mm[ω_, l_, m_, r_, θ_] := - ϵg/8 * ξbar^2/ξ^2 * (Dnadjbar[0, ω, m,
       Dnadjbar[0, ω, m, Δ[r]^2 * Rbar[2, ω, l, m, r]] * Sbar[2, ω, l, m, θ]] -
      2/ξbar * Dnadjbar[0, ω, m, Δ[r]^2 * Rbar[2, ω, l, m, r] * Sbar[2, ω, l, m, θ]]) /. ωsub
```

We export $h_{\mu\nu}^{\mathrm{ORG}}$ in terms of $R_{\omega l m}^{(+2)}(r)$ and $S_{\omega l m}^{(+2)}(\theta)$.

In[ ]:=
```
hORGBLExp = hORGBL[t, r, θ, ϕ];
hORGInExp = hORGIn[t, r, θ, ϕ];
hORGOutExp = hORGOut[t, r, θ, ϕ];
DumpSave["./metric-perturbations/hORGExpressions.mx",
  {hORGBLExp, hORGInExp, hORGOutExp}];
```

We export $h_{a\,b}^{\text{ORG}}$ in terms of $R_{\omega l\,m}^{(+2)}(r)$ and $S_{\omega l\,m}^{(+2)}(\theta)$.

In[ ]:=
```
starttime = AbsoluteTime[];
{
  {hll[t, r, θ, ϕ], 0, hlm[t, r, θ, ϕ], hlmbar[t, r, θ, ϕ]},
  {0, 0, 0, 0},
  {hlm[t, r, θ, ϕ], 0, hmm[t, r, θ, ϕ], 0},
  {hlmbar[t, r, θ, ϕ], 0, 0, hmbarmbar[t, r, θ, ϕ]}
 };
habORG = Collect[%, {e^(-i m ϕ+i t ωbar), e^(i m ϕ-i*ω*t)}, Collect[#,
     {A, B, Abar, Bbar}, Collect[#, Rlist, Collect[#, Slist, Simplify] &] &] &];
DumpSave["./metric-perturbations/habORG.mx", habORG];
AbsoluteTime[] - starttime
```

Out[ ]=
```
45.005304
```